Cubic Splines

James Keesling

1 Definition of Cubic Spline

Given a function $f(x)$ defined on an interval $[a, b]$ we want to fit a curve through the points $\{(x_0, f(x_0)), (x_1, f(x_1)), \ldots, (x_n, f(x_n))\}$ as an approximation of the function $f(x)$. We assume that the points are given in order $a = x_0 < x_1 < x_2 < \cdots < x_n = b$ and let $h_i = x_{i+1} - x_i$. The method of approximation we describe is called *cubic spline interpolation*. The cubic spline is a function $S(x)$ on $[a, b]$ with the following properties.

$S(x)\big|_{[x_i, x_{i+1}]} = S_i(x)$ is a cubic polynomial for $i = 0, 1, 2, \ldots, n-1$
$S_i(x_i) = f(x_i)$ for $i = 0, 1, 2, \ldots, n-1$
$S_i(x_{i+1}) = f(x_{i+1})$ for $i = 0, 1, 2, \ldots, n-1$
$S_i'(x_{i+1}) = S_{i+1}'(x_{i+1})$ for $i = 0, 1, 2, \ldots, n-2$
$S_i''(x_{i+1}) = S_{i+1}''(x_{i+1})$ for $i = 0, 1, 2, \ldots, n-2$
$S_0''(x_0) = S_{n-1}''(x_n)$ [*free boundary condition*] or
$S_0'(x_0) = f'(x_0)$ and $S_{n-1}'(x_n) = f'(x_n)$ [*clamped boundary condition*]

2 Determining the Coefficients of the Cubic Polynomials

Since each $S_i(x) = a_i + b_i \cdot (x - x_i) + c_i \cdot (x - x_i)^2 + d_i \cdot (x - x_i)^3$ has four constants to be determined, we have $4n$ unknowns and the above conditions give us $4n$ equations. For the free boundary case we can simplify the solutions of the equations to the following.

$a_i = f(x_i)$ for $i = 0, 1, 2, \ldots, n-1$ and define $a_n = f(x_n)$

$Ax = b$ where

$$A = \begin{bmatrix} 1 & 0 & 0 & 0 & \cdots & 0 \\ h_0 & 2(h_0 + h_1) & h_1 & 0 & \cdots & 0 \\ 0 & h_1 & 2(h_1 + h_2) & h_2 & \cdots & 0 \\ 0 & 0 & h_2 & 2(h_2 + h_3) & \cdots & 0 \\ 0 & 0 & 0 & h_3 & \ddots & \\ \vdots & \vdots & & & & \\ 0 & 0 & 0 & \cdots & 0 & 1 \end{bmatrix}$$

$$b = \begin{bmatrix} 0 \\ \frac{3}{h_1}(a_2 - a_1) - \frac{3}{h_0}(a_1 - a_0) \\ \vdots \\ \frac{3}{h_{n-1}}(a_n - a_{n-1}) - \frac{3}{h_{n-2}}(a_{n-1} - a_{n-2}) \\ 0 \end{bmatrix}$$

and

$$x = \begin{bmatrix} c_0 \\ c_1 \\ c_2 \\ \vdots \\ c_n \end{bmatrix}$$

We also have that $d_i = \frac{c_{i+1} - c_i}{3h_i}$ for $i = 0, 1, 2, \ldots, n-1$ and $b_i = \frac{1}{h_i}(a_{i+1} - a_i) - \frac{h_i}{3}(2c_i + c_{i+1})$ for $i = 0, 1, 2, \ldots, n-1$.

3 A Program to Find the Coefficients

The following TI-89 program will determine the coefficients of $S_0(x), S_1(x), \ldots, S_{n-1}(x)$. The input is the vector of x_i-values, $xvec$, and the vector of $f(x_i)-$values (or y_i-values), $yvec$. The variable n is the length of these vectors less one. The output is the $n \times 4$ matrix $coef$. The i^{th} row gives the coefficients $\{a_{i-1}, b_{i-1}, c_{i-1}, d_{i-1}\}$ of $S_{i-1}(x) = a_{i-1} + b_{i-1}(x - x_{i-1}) + c_{i-1}(x - x_{i-1})^2 + d_{i-1}(x - x_{i-1})^3$.

```
:cubsplin(xvec, yvec, n)
: Prgm
:newMat(n,4) → coef
:newMat(1,n) → h
:newMat(n+1,n+1) → temp1
:newMat(n+1,1) → temp2
:For i,1,n
:xvec[1,i+1]-xvec[1,i] → h[1,i]
:EndFor
:For i,1,n
:yvec[1,i] → coef[i,1]
:EndFor
:For i,1,n-1
:h[1,i] → temp1[i+1,i]
:2*(h[1,i] + h[1,i+1]) → temp1[i+1,i+1]
:h[1,i+1] → temp1[i+1,i+2]
```

```
:EndFor
:1 → temp1[1,1]
:1 → temp1[n+1,n+1]
:For i,1,n-1
:3/(h[1,i+1])*(yvec[1,i+2]-yvec[1,i+1])-3/(h[1,i])*(yvec[1,i+1]-yvec[1,i]) → temp2[i+1,1]
:EndFor
:temp1∧(-1) * temp2 → temp2
:For i,1,n
:temp2[i,1] → coef[i,3]
:EndFor
:For i,1,n
:(temp2[i+1,1] - temp2[i,1])/(3*h[1,i]) → coef[i,4]
:1/(h[1,i])*(yvec[1,i+1]-yvec[1,i]) - h[1,i]/3*(2*temp2[i,1] + temp2[i+1,1]) → coef[i,2]
:EndFor
:EndPrgm
```

To be sure that you program is working try the following example.

Example 3.1. *Determine the coefficients of the cubic spline through the following points* $\{(0,0),(1,1),(2,8),\left(\frac{5}{2},9\right)\}$.

The coefficient matrix $coef$ should be the following.

$$\begin{bmatrix} 0 & -\frac{12}{11} & 0 & \frac{23}{11} \\ 1 & \frac{57}{11} & \frac{69}{11} & -\frac{49}{11} \\ 8 & \frac{48}{11} & -\frac{78}{11} & \frac{52}{11} \end{bmatrix}$$

This says that the spline is given by the following formula.

$$S(x) = \begin{cases} -\frac{12}{11}x + \frac{23}{11}x^3 & 0 \le x \le 1 \\ 1 + \frac{57}{11}(x-1) + \frac{69}{11}(x-1)^2 - \frac{49}{11}(x-1)^3 & 1 \le x \le 2 \\ 8 + \frac{48}{11}(x-2) - \frac{78}{11}(x-2)^2 + \frac{52}{11}(x-2)^3 & 2 \le x \le \frac{5}{2} \end{cases}$$